Pensieve header: A unified verification notebook for the sl_2-portfolio project; continues pensieve://Projects/PPSA/nb/Verification.pdf.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Portfolio"];
NotebookOpen[wdir <> "\\MakeSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization / Utilities

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$p = 3; $k = 1; ϵ /: ϵ^k_· /; k > $k := 0;
(* $k can't be ∞ at least because of Quesne. Can't be ≤
 1 at least because of the explicit ϵ² in S𝔻$g. *)
SetAttributes[{SS, SST}, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
   Collect[Normal@Series[ℰ, {ℏ, 0, $p}], ℏ, Together] ];
SST[ℰ_] := Block[{ℏ, ϵ},
   Collect[Normal@Series[ℰ /. {T_i_ → e^(ℏ t_i/2), T → e^(ℏ t/2)}, {ℏ, 0, $p}], ℏ, Together] ];
Simp[ℰ_, op_] := Collect[ℰ, _CU | _QU, op];
Simp[ℰ_] := Simp[ℰ, Collect[Normal@Series[#, {ℏ, 0, $p}], ℏ, Expand] &];
SimpT[ℰ_] := Collect[ℰ, _CU | _QU,
   Collect[Normal@Series[# /. {T_i_ → e^(ℏ t_i/2), T → e^(ℏ t/2)}, {ℏ, 0, $p}], ℏ, Expand] &];
```

Differential polynomials (DP):

Utils

```
DP_{α_→D_x_, β_→D_y_}[P_][λ_] :=
 Total[CoefficientRules[P, {α, β}] /. ({m_, n_} → c_) :> c D[λ, {x, m}, {y, n}]]
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, CE, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts} },
   (#_U = U@#) & /@ gs;
   gp = Alternatives @@ gs; gp = gp | gp_; (* gen's pattern *)
   sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
   cp = Alternatives @@ cs; (* cent's pattern *)
   CE[ℰ_] := Collect[ℰ, _U, (Expand[#] /. ℏ^d_ /; d > $p :> 0) &];
   U_{i_}[ℰ_] := ℰ /. {t : cp :> t_i, u_U :> Replace[u, x_ :> x_i, 1]};
   U_{i_}[NCM[]] = pow[ℰ_, 0] = U@{} = 1_U = U[];
   B[U@(x_)_{i_}, U@(y_)_{i_}] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
   B[U@(x_)_{i_}, U@(y_)_{j_}] /; i =!= j := 0;
   B[U@y_, U@x_] := CE[-B[U@x, U@y]];
   x_ ** 1_U := x; 1_U ** x_ := x;
   (a_. * x_U) ** (b_. * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
   U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
     U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
   U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
   U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
   U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
   U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
   U@{l_, r___} := U@{Expand[l], r};
   U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
   𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
     sp = Replace[{specs}, l_List :> l_null, {1}];
     vs = Join @@ (First /@ sp);
     us = Join @@ (sp /. l_{s_} :> (l /. x_{i_} :> x_s));
     CE[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :> c U@(us^p)
       ]] /. x_null :> x
    ];
   pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
   𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) :> c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
   S_{i_}[c_. * u_U] := CE[(c /. S_i[U, Centrals]) DeleteCases[u, _i] **
      U_i[NCM @@ Reverse@Cases[u, x_i :> S@U@x]]]; ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
   Replace[ongs, (g_ → img_) :> (m[U[g]] = img), {1}];
   m[1_U] = 1_V;
   m[U[g_{i_}]] := V_i[m[U@g]];
   m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
   m[ℰ_] := Simp[ℰ /. oncs /. u_U :> m[u]]; )
```

Meta-Operations

QLImplementation

```
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[a_CU, y_CU] = -γ y_CU; B[x_CU, a_CU] = -γ x_CU;
B[x_CU, y_CU] = 2 ϵ a_CU - t 1_CU;
(S@CU@y = -y_CU; S@a_CU = -a_CU; S@x_CU = -x_CU;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.921875, {(28 t^2 γ^4 + 116 t γ^5 ϵ + 120 γ^6 ϵ^2) CU[y, y, y, x, x] + (4 t^3 γ + 8 t^2 γ^2 ϵ + 4 t γ^3 ϵ^2)
    CU[y, y, a, a, a, x] + <<22>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Verifying that S is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y_1, a_1, x_1}},
 Table[HL@Simp[S_1[z1 ** z2] - S_1[z2] ** S_1[z1]],
  {z1, bas}, {z2, bas} ] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of S on products of triples:

```
With[{bas = CU /@ {y_1, a_1, x_1}},
 Table[HL@Simp[z1 ** z2 ** z3 - S_1@S_1[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle/\mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ℏ)]; (*T=SS[e^(ℏ t/2)];*)
B[a_QU, y_QU] = -γ y_QU; B[x_QU, a_QU] = -γ QU@x;
B[x_QU, y_QU] = (q - 1) QU@{y, x} + O_QU[SS[(1 - T^2 e^(-2 ϵ a ℏ)) / ℏ], {a}];
(S@y_QU = O_QU[SS[-T^-2 e^(ℏ ϵ a) y], {a, y}]; S@a_QU = -a_QU; S@x_QU = O_QU[SS[-e^(ℏ ϵ a) x], {a, x}];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to \gamma\, QU[y], \{QU[y], QU[x]\} \to \frac{(-1 + T^2)\, QU[\,]}{\hbar} - 2\, T^2 \epsilon\, QU[a] + 2\, T^2 \epsilon^2 \hbar\, QU[a, a] + \left(-\gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x]\},$
$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y], \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$
$\{\{QU[x], QU[y]\} \to \frac{(1 - T^2)\, QU[\,]}{\hbar} + 2\, T^2 \epsilon\, QU[a] - 2\, T^2 \epsilon^2 \hbar\, QU[a, a] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x],$
$\{QU[x], QU[a]\} \to -\gamma\, QU[x], \{QU[x], QU[x]\} \to 0\}\}$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}

Verifying associativity on a “random” triple (~34 secs @ $p=5, $k=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp]
  }] // Timing
```

$\{40.2969,$
$\{\left(135\, \gamma^6 \epsilon^2 - 730\, T^2 \gamma^6 \epsilon^2 + 715\, T^4 \gamma^6 \epsilon^2 + \frac{28\, \gamma^4 - 56\, T^2 \gamma^4 + 28\, T^4 \gamma^4}{\hbar^2} + \frac{82\, \gamma^5 \epsilon - 280\, T^2 \gamma^5 \epsilon + 198\, T^4 \gamma^5 \epsilon}{\hbar}\right)$
$QU[y, y, y, x, x] + \ll 22 \gg +$
$(1 + 8\, \gamma \epsilon \hbar + 32\, \gamma^2 \epsilon^2 \hbar^2)\, QU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0\}\}$

Verifying that S is an anti-homomorphism on QU:

```
With[{bas = QU /@ {y1, a1, x1}},
 Table[{z1, z2} → HL@Simp[S1[z1 ** z2] - S1[z2] ** S1[z1]],
  {z1, bas}, {z2, bas} ] ]
```

$\{\{\{QU[y_1], QU[y_1]\} \to 0, \{QU[y_1], QU[a_1]\} \to 0, \{QU[y_1], QU[x_1]\} \to 0\},$
$\{\{QU[a_1], QU[y_1]\} \to 0, \{QU[a_1], QU[a_1]\} \to 0, \{QU[a_1], QU[x_1]\} \to 0\},$
$\{\{QU[x_1], QU[y_1]\} \to 0, \{QU[x_1], QU[a_1]\} \to 0, \{QU[x_1], QU[x_1]\} \to 0\}\}$

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs @ $p=5, $k=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
  Short[lhs = z1 ** (z2 ** z3)],
  Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
  Expand[Limit[rhs /. {QU → CU, T → ⅇ^(ℏ t/2)}, ℏ → 0] - lhs] // HL
 }] // Timing
```

```
{33.5625, {2 (8 t^2 γ^4 + 16 t γ^5 ∈) CU[y, y, y, x, x] +
   (8 t γ^5 ∈ + 16 γ^6 ∈^2) CU[y, y, y, x, x] + <<106>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x],
  (-8 T^2 γ^6 ∈^2 + 8 T^4 γ^6 ∈^2) QU[y, y, y, x, x] + (-16 T^2 γ^6 ∈^2 + 16 T^4 γ^6 ∈^2) QU[y, y, y, x, x] +
   <<488>> + (γ ∈ ℏ + 15/2 γ^2 ∈^2 ℏ^2) QU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -x_CU, a → -a_CU, x → -y_CU}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ∈ a) x], {a, x}],
  a → -a_QU, x → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ∈ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

```
{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}
```

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\left\{QU[y] \to -\frac{QU[x]}{T} - \frac{\epsilon\hbar\, QU[a,x]}{T} - \frac{\epsilon^2\hbar^2\, QU[a,a,x]}{2T} \to QU[y],\ QU[a] \to -QU[a] \to QU[a],\right.$$
$$\left. QU[x] \to \left(-\frac{1}{T} + \frac{\gamma\epsilon\hbar}{T} - \frac{\gamma^2\epsilon^2\hbar^2}{2T}\right) QU[y] + \left(-\frac{\epsilon\hbar}{T} + \frac{\gamma\epsilon^2\hbar^2}{T}\right) QU[y,a] - \frac{\epsilon^2\hbar^2\, QU[y,a,a]}{2T} \to QU[x]\right\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

```
A𝔻$f = γ ((Cosh[ℏ (a ϵ + γ ϵ/2 - t/2)] - Cosh[ℏ Sqrt[((t - γ ϵ)/2)^2 + ϵ ω]]) /
    (ℏ ⅇ^(ℏ ((a+γ) ϵ-t/2)) Sinh[γ ϵ ℏ/2] (a^2 ϵ + a γ ϵ - a t - ω)));
```

Scaling behaviour of A𝔻$f:

```
HL@Simplify[A𝔻$f == ((A𝔻$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
True
```

```
HL@FullSimplify[
  A𝔻$f == ((A𝔻$f /. γ → 1) /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
True
```

ADeq

```
A𝔻$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
 {a → a_CU, x → CU@x, y → 𝕊_CU[SS[A𝔻$f], a → a_CU, ω → A𝔻$ω] ** y_CU}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[SimpT[A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2]]], {z1, bas}, {z2, bas}] ]
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

```
S𝔻$g = Sqrt[((2 γ (Cosh[ℏ/2 Sqrt[t^2 + γ^2 ϵ^2 + 4 ϵ ϖ]] - Cosh[(t - ϵ γ - 2 ϵ a)/(2 / ℏ)])) /
    (Sinh[γ ϵ ℏ/2] (t (2 a + γ) - 2 a (a + γ) ϵ + 2 ϖ) ℏ))];
```

Verify agreement with the formulas in pensieve://People/VanDerVeen/Dequant1.pdf:

```
{SD$P = (Cosh[ħ ((ϵ-t)/2 + ϵ a)] - Cosh[ħ Sqrt[(t^2+ϵ^2)/4 + ϵ ϖ]]) / (ħ Sinh[-ϵ ħ/2] (ϖ - ϵ a^2 + (t - ϵ) a + t/2)),
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
    SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 FullSimplify[SD$g == FullSimplify[
    Sqrt[SD$Q] /. c → a + 1/2 /. {ħ → γ^2 ħ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}(t^2+\epsilon^2)+\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,(t-\epsilon)-a^2\epsilon+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},\right.$$

$$\left.-\left(\left(4\left(\mathrm{Cosh}\left[\frac{1}{2}(t-2\,c\,\epsilon)\,\hbar\right]-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\epsilon+4\,\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True}\right\}$$

SDeq

```
SD$f = FullSimplify[e^(ħ (t/2-ϵ a)) (SD$g /. {a → -a, t → -t})];
```

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ 1_CU / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → a_CU,
  x → S_CU[SS[SD$f], a → a_CU, ϖ → SD$ϖ] ** x_CU,
  y → S_CU[SS[SD$g], a → a_CU, ϖ → SD$ϖ] ** y_CU }]
```

Verifying the θ-symmetry:

```
Table[HL@SimpT[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

{True, True, True}

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@SimpT[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

R in QU.

Quesne's formula:

Quesne

```
𝕖_{q_,k_}[x_] := 𝕖^(Sum_{j=1}^{k} ((1 - q)^j x^j)/(j (1 - q^j))); 𝕖_{q_}[x_] := 𝕖_{q,$k}[x]
```

```
Table[Together@SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)}, 1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

{1, 1, 1, 1, 1, 1}

R

```
QU[R_{i_,j_}] := 𝕆_QU[SS[𝕖^(ℏ b1 a2) 𝕖_q[ℏ y1 x2] /. b1 → γ^-1 (ϵ a1 - t_i)], {y1, a1}_i, {a2, x2}_j];
QU[R^{-1}_{i_,j_}] := S_j@QU[R_{i,j}];
```

```
QU[R_{3,4}] // Short
```

$\text{QU}[\,] + \frac{\epsilon\hbar\,\text{QU}[a_3, a_4]}{\gamma} + \hbar\,\text{QU}[y_3, x_4] + \frac{\epsilon^2\hbar^2\,\text{QU}[a_3, a_3, a_4, a_4]}{2\gamma^2} + \ll 18 \gg + \frac{\epsilon\hbar^3\,\text{QU}[a_3, a_4, a_4, a_4]\,t_3^2}{2\gamma^3} + \frac{\hbar^3\,\text{QU}[y_3, a_4, a_4, x_4]\,t_3^2}{2\gamma^2} - \frac{\hbar^3\,\text{QU}[a_4, a_4, a_4]\,t_3^3}{6\gamma^3}$

Verifying R2 (~2 secs @ $p=4, $k=2):

```
QU[R_{1,2} ** R^{-1}_{1,2}] // Simp // HL // Timing
```

{0.5625, QU[]}

Verifying R3 (~156 secs @ $p=4, $k=2):

```
{Short[lhs = QU[R_{1,2} ** R_{1,3} ** R_{2,3}]], HL@SimpT[lhs - QU[R_{2,3} ** R_{1,3} ** R_{1,2}]]} // Timing
```

$\Big\{11.375, \Big\{\text{QU}[\,] + \frac{\epsilon\hbar\,\text{QU}[a_1, a_2]}{\gamma} + \frac{\epsilon\hbar\,\text{QU}[a_1, a_3]}{\gamma} + \ll 453 \gg + \text{QU}[y_1, y_1, a_3, x_3, x_3]\left(-\frac{\hbar^3 t_2}{2\gamma} + \frac{\hbar^3 t_2 T_2^2}{\gamma} - \frac{\hbar^3 t_2 T_2^4}{2\gamma}\right) + \text{QU}[y_1, y_1, y_1, x_3, x_3, x_3]\left(\frac{\hbar^3}{6} - \frac{1}{2}\hbar^3 T_2^2 + \frac{1}{2}\hbar^3 T_2^4 - \frac{1}{6}\hbar^3 T_2^6\right), 0\Big\}\Big\}$

The representation ρ

rho

```
ρ@y_CU = ρ@y_QU = (0 0; ϵ 0); ρ@a_CU = ρ@a_QU = (γ 0; 0 0);
ρ@x_CU = (0 γ; 0 0); ρ@x_QU = SS@(0 (1 - e^(-γ ϵ ℏ)) / (ϵ ℏ); 0 0);
ρ[e^𝓔_] := MatrixExp[ρ[𝓔]];
ρ[𝓔_] :=
  (𝓔 /. {t → γ ϵ, T → e^(ℏ γ ϵ/2)} /. (U : CU | QU)[u___] :→ Fold[Dot, (1 0; 0 1), ρ /@ U /@ {u}])
```

Verifying that ρ represents CU and QU:

```
Table[ρ[z1 ** z2] == ρ[z1].ρ[z2] // SS // HL,
 {U, {CU, QU}}, {z1, U /@ {y, a, x}}, {z2, U /@ {y, a, x}} ]
```

```
{{{True, True, True}, {True, True, True}, {True, True, True}},
 {{True, True, True}, {True, True, True}, {True, True, True}}}
```

The Classical Logos CΛ

Lemma 3C. To degree k,
$\mathbb{O}_{CU}\left(e^{\eta y+\xi x+\delta y x} \mid x y\right) = \mathbb{O}_{CU}\left(v\, e^{v(-t \xi \eta+\eta y+\xi x+\delta y x)}\, C\Lambda_k(\epsilon, \gamma, y, a, x, \eta, \xi, \delta) \mid y a x\right)$, with $v = (1 + t\,\delta)^{-1}$ and where $C\Lambda_k(\epsilon, \gamma, y, a, x, \eta, \xi, \delta)$ is a fixed polynomial of degree at most $4k$ in y, $\sqrt{a}$, x, η, ξ, with scalar coefficients.

Comment. Even better, $\log(C\Lambda_k)$ is of degree at most $2k+2$ in said variables.

```
eqn = ρ[e^(ξ x_CU)].ρ[e^(η y_CU)] == ρ[e^(d y_CU)].ρ[e^(c (t 1_CU - 2 ϵ a_CU))].ρ[e^(b x_CU)]
```

```
{{1 + γ ϵ η ξ, γ ξ}, {ϵ η, 1}} == {{e^(-c γ ϵ), b e^(-c γ ϵ) γ}, {d e^(-c γ ϵ) ϵ, e^(c γ ϵ) + b d e^(-c γ ϵ) γ ϵ}}
```

```
sol = Solve[Thread[Flatten /@ eqn], {d, b, c}][[1]] /. C[1] → 0
```

$$\left\{d \to \frac{\eta}{1+\gamma \epsilon \eta \xi},\ b \to \frac{\xi}{1+\gamma \epsilon \eta \xi},\ c \to \frac{\mathrm{Log}\left[\frac{1}{1+\gamma \epsilon \eta \xi}\right]}{\gamma \epsilon}\right\}$$

Proof of Lemma 3C. We know that $\mathbb{O}_{CU}\left(e^{\xi x+\eta y} \mid x y\right) = \mathbb{O}_{CU}\left(e^{c t+a y-2\epsilon c a+b x} \mid y a x\right)$, with $\left\{d \to \frac{\eta}{1+\gamma\epsilon\eta\xi}, b \to \frac{\xi}{1+\gamma\epsilon\eta\xi}, c \to \frac{\mathrm{Log}[1+\gamma\epsilon\eta\xi]}{-\gamma\epsilon}\right\}$. Expanding in ϵ we get
$\mathbb{O}_{CU}\left(e^{\xi x+\eta y} \mid x y\right) = \mathbb{O}_{CU}\left(\lambda_\epsilon(\xi, \eta)\, e^{\eta y+\xi x-\eta \xi t} \mid y a x\right) = \mathbb{O}_{CU}\left(\lambda_\epsilon(\partial_x, \partial_y)\, e^{\eta y+\xi x-\eta\xi t} \mid y a x\right)$ and so
$\mathbb{O}_{CU}\left(e^{\eta y+\xi x+\delta y x} \mid x y\right) = \mathbb{O}\left(\lambda_\epsilon(\partial_x, \partial_y)\, e^{\delta \partial_\eta \partial_\xi}\, e^{\eta y+\xi x-\eta\xi t} \mid y a x\right) = \mathbb{O}\left(\lambda_\epsilon(\partial_x, \partial_y)\, v\, e^{v(-t\xi\eta+\eta y+\xi x+\delta y x)} \mid y a x\right)$.

Logos

```
SSϵ[ℰ_] := Block[{ϵ}, Collect[Normal@Series[ℰ, {ϵ, 0, $k}], ϵ, Together]];
(* Shielded ϵ-Series *)
CΛ[t1_, y1_, a1_, x1_, ξ1_, η1_, δ_] := Module[
  {eqn, d, b, c, sol, λ, q, ν, ξ, η},
  eqn = ρ[ⅇ^(ξ x_CU)].ρ[ⅇ^(η y_CU)] == ρ[ⅇ^(d y_CU)].ρ[ⅇ^(c (t 1_CU - 2 ϵ a_CU))].ρ[ⅇ^(b x_CU)];
  sol = Solve[Thread[Flatten /@ eqn], {d, b, c}][[1]] /. C[1] → 0;
  λ = Simplify[ⅇ^(-η y-ξ x+η ξ t) SSϵ[ⅇ^(c t + d y - 2 ϵ c a + b x) /. sol]];
  q = ⅇ^(ν (-t ξ η+η y+ξ x+δ y x));
  Collect[ν q^-1 DP_(ξ→D_x,η→D_y)[λ][q] /. ν → (1 + t δ)^-1, ϵ, Simplify] /.
   {t → t1, y → y1, a → a1, x → x1, ξ → ξ1, η → η1} ];
```

CΛ[t, y, a, x, ξ, η, δ]

$$\frac{1}{1+t\delta}+\frac{1}{24\left(1+t\delta\right)^9}$$
$$\epsilon^2\Big(48a^2\left(1+t\delta\right)^4\left(2\delta^2\left(1+t\delta\right)^2+4\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)-$$
$$24a\gamma\left(1+t\delta\right)^4\left(2\delta^2\left(1+t\delta\right)^2+4\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)-$$
$$48ay\gamma\left(1+t\delta\right)^3(x\delta+\eta)$$
$$\left(6\delta^2\left(1+t\delta\right)^2+6\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)+24y\gamma^2\left(1+t\delta\right)^3$$
$$(x\delta+\eta)\left(6\delta^2\left(1+t\delta\right)^2+6\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)-48ax\gamma$$
$$\left(1+t\delta\right)^3(y\delta+\xi)\left(6\delta^2\left(1+t\delta\right)^2+6\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)+$$
$$24x\gamma^2\left(1+t\delta\right)^3(y\delta+\xi)\Big(6\delta^2\left(1+t\delta\right)^2+6\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+$$
$$(x\delta+\eta)^2(y\delta+\xi)^2\Big)+12y^2\gamma^2\left(1+t\delta\right)^2(x\delta+\eta)^2$$
$$\left(12\delta^2\left(1+t\delta\right)^2+8\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)+12x^2\gamma^2$$
$$\left(1+t\delta\right)^2(y\delta+\xi)^2\left(12\delta^2\left(1+t\delta\right)^2+8\delta\left(1+t\delta\right)(x\delta+\eta)(y\delta+\xi)+(x\delta+\eta)^2(y\delta+\xi)^2\right)+$$
$$24at\gamma\left(1+t\delta\right)^2\Big(6\delta^3\left(1+t\delta\right)^3+18\delta^2\left(1+t\delta\right)^2(x\delta+\eta)(y\delta+\xi)+$$
$$9\delta\left(1+t\delta\right)(x\delta+\eta)^2(y\delta+\xi)^2+(x\delta+\eta)^3(y\delta+\xi)^3\Big)-$$
$$8t\left(\gamma+t\gamma\delta\right)^2\Big(6\delta^3\left(1+t\delta\right)^3+18\delta^2\left(1+t\delta\right)^2(x\delta+\eta)(y\delta+\xi)+$$
$$9\delta\left(1+t\delta\right)(x\delta+\eta)^2(y\delta+\xi)^2+(x\delta+\eta)^3(y\delta+\xi)^3\Big)+$$
$$24xy\left(\gamma+t\gamma\delta\right)^2\Big(6\delta^3\left(1+t\delta\right)^3+18\delta^2\left(1+t\delta\right)^2(x\delta+\eta)(y\delta+\xi)+$$
$$9\delta\left(1+t\delta\right)(x\delta+\eta)^2(y\delta+\xi)^2+(x\delta+\eta)^3(y\delta+\xi)^3\Big)-$$
$$12ty\gamma^2\left(1+t\delta\right)(x\delta+\eta)\Big(24\delta^3\left(1+t\delta\right)^3+36\delta^2\left(1+t\delta\right)^2(x\delta+\eta)(y\delta+\xi)+$$
$$12\delta\left(1+t\delta\right)(x\delta+\eta)^2(y\delta+\xi)^2+(x\delta+\eta)^3(y\delta+\xi)^3\Big)-$$
$$12tx\gamma^2\left(1+t\delta\right)(y\delta+\xi)\Big(24\delta^3\left(1+t\delta\right)^3+36\delta^2\left(1+t\delta\right)^2(x\delta+\eta)(y\delta+\xi)+$$
$$12\delta\left(1+t\delta\right)(x\delta+\eta)^2(y\delta+\xi)^2+(x\delta+\eta)^3(y\delta+\xi)^3\Big)+$$
$$3t^2\gamma^2\Big(24\delta^4\left(1+t\delta\right)^4+96\delta^3\left(1+t\delta\right)^3(x\delta+\eta)(y\delta+\xi)+72\delta^2\left(1+t\delta\right)^2$$
$$(x\delta+\eta)^2(y\delta+\xi)^2+16\delta\left(1+t\delta\right)(x\delta+\eta)^3(y\delta+\xi)^3+(x\delta+\eta)^4(y\delta+\xi)^4\Big)\Big)+$$
$$\frac{1}{2\left(1+t\delta\right)^5}\epsilon\Big(4a\left(1+t\delta\right)^2\left(\left(t+xy\right)\delta^2+\eta\xi+\delta\left(1+y\eta+x\xi\right)\right)+$$
$$\gamma\Big(2t^3\delta^4+4t^2\delta^2\left(\delta-xy\delta^2+\eta\xi\right)-2\Big(y\eta\left(\delta\left(2+y\eta\right)+\eta\xi\right)+x^2\delta\left(2y^2\delta^2+3y\delta\xi+\xi^2\right)+$$
$$x\left(3y^2\delta^2\eta+4y\delta(\delta+\eta\xi)+\xi\left(2\delta+\eta\xi\right)\right)\Big)-t\Big(3x^2y^2\delta^4-4\delta\eta\xi-\eta^2\xi^2+$$
$$4xy\delta^3\left(3+y\eta+x\xi\right)+\delta^2\left(-2+y^2\eta^2+4x\xi+x^2\xi^2+4y(\eta+x\eta\xi)\right)\Big)\Big)\Big)$$

```
{Short[lhs = 𝕆CU[SS[𝕖^(ℏ (ξ x+η y+δ x y))], {x, y}], 5], HL[lhs ==
   𝕆CU[SS[𝕖^(ℏ ν (ξ x+η y+δ x y-t ℏ ξ η)) CΛ[t, y, a, x, ℏ ξ, ℏ η, ℏ δ] /. ν → (1 + ℏ t δ)^-1], {y, a, x}]]}
```

$\{(1 - t\delta\hbar + t^2\delta^2\hbar^2 + t\gamma\delta^2\epsilon\hbar^2 - t\eta\xi\hbar^2 - t^3\delta^3\hbar^3 - 3t^2\gamma\delta^3\epsilon\hbar^3 - 2t\gamma^2\delta^3\epsilon^2\hbar^3 + 2t^2\delta\eta\xi\hbar^3 + 2t\gamma\delta\epsilon\eta\xi\hbar^3)\,\mathrm{CU}[\,] +$
$(2\delta\epsilon\hbar - 4t\delta^2\epsilon\hbar^2 - 2\gamma\delta^2\epsilon^2\hbar^2 + 2\epsilon\eta\xi\hbar^2 + 6t^2\delta^3\epsilon\hbar^3 + 12t\gamma\delta^3\epsilon^2\hbar^3 - 8t\delta\epsilon\eta\xi\hbar^3 - 4\gamma\delta\epsilon^2\eta\xi\hbar^3)\,\mathrm{CU}[a] +$
$(\xi\hbar - 2t\delta\xi\hbar^2 - 2\gamma\delta\epsilon\xi\hbar^2 + 3t^2\delta^2\xi\hbar^3 + 9t\gamma\delta^2\epsilon\xi\hbar^3 + 6\gamma^2\delta^2\epsilon^2\xi\hbar^3 - t\eta\xi^2\hbar^3 - \gamma\epsilon\eta\xi^2\hbar^3)\,\mathrm{CU}[x] +$
$(\eta\hbar - 2t\delta\eta\hbar^2 - 2\gamma\delta\epsilon\eta\hbar^2 + 3t^2\delta^2\eta\hbar^3 + 9t\gamma\delta^2\epsilon\eta\hbar^3 + 6\gamma^2\delta^2\epsilon^2\eta\hbar^3 - t\eta^2\xi\hbar^3 - \gamma\epsilon\eta^2\xi\hbar^3)\,\mathrm{CU}[y] + (4\delta^2\epsilon^2\hbar^2 - 12t\delta^3\epsilon^2\hbar^3 + 8\delta\epsilon^2\eta\xi\hbar^3)\,\mathrm{CU}[a,a] +$
$(4\delta\epsilon\xi\hbar^2 - 12t\delta^2\epsilon\xi\hbar^3 - 18\gamma\delta^2\epsilon^2\xi\hbar^3 + 2\epsilon\eta\xi^2\hbar^3)\,\mathrm{CU}[a,x] + \ll 18 \gg +$
$\frac{1}{2}\delta\eta^2\hbar^3\,\mathrm{CU}[y,y,y,x] + 3\delta^3\epsilon\hbar^3\,\mathrm{CU}[y,y,a,x,x] + \frac{1}{2}\delta^2\xi\hbar^3\,\mathrm{CU}[y,y,x,x,x] +$
$\frac{1}{2}\delta^2\eta\hbar^3\,\mathrm{CU}[y,y,y,x,x] + \frac{1}{6}\delta^3\hbar^3\,\mathrm{CU}[y,y,y,x,x,x],\ \mathrm{True}\}$

The Quantum Logos QΛ

Goal 1: In QU, compute $F = e^{-\eta y} e^{\xi x} e^{\eta y} e^{-\xi x}$.

First compute $G = e^{\xi x}\, y\, e^{-\xi x}$, a finite sum.

Now F satisfies the ODE $\partial_\eta F = \partial_\eta\left(e^{-\eta y} e^{\eta G}\right) = -yF + FG$ with initial conditions $F(\eta = 0) = 1$. So we set it up and solve:

Logos

```
QΛ[T1_, y1_, a1_, x1_, ξ1_, η1_, δ_] := Module[
   {G, F, fs, f, bs, e, b, es, λ, q, ν, ξ, η, t},
   G = Simp[
     Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[x_QU ** # - # ** x_QU] &, y_QU, $k + 1]];
   fs = Flatten@Table[f_{l,i,j,k}[η], {l, 0, $k}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}];
   F = fs.(bs = fs /. f_{l_,i_,j_,k_}[η] :→ ϵ^l QU@{y^i, a^j, x^k});
   es = Flatten[
     Table[Coefficient[e, b] == 0, {e, {F - 1_QU /. η → 0, F ** G - y_QU ** F - ∂_η F}}, {b, bs}]];
   {λ} = F /. DSolve[es, fs, η] /. {𝕖- → 1, QU → Times};
   q = 𝕖^(ν (-t ξ η+η y+ξ x+δ y x));
   Collect[ν q^-1 DP_{ξ→D_x, η→D_y}[λ][q] /. ν → (1 + t δ)^-1 /. t → (T^2 - 1) / ℏ, ϵ, Simplify] /.
    {T → T1, y → y1, a → a1, x → x1, ξ → ξ1, η → η1} ];
```

```
QΛ[T, y, a, x, ξ, η, δ]
```

$$\frac{\hbar}{\left(-1+T^2\right)\delta+\hbar}+\frac{1}{4\left(\left(-1+T^2\right)\delta+\hbar\right)^5}\epsilon\hbar^2\Big(8aT^2\left(\left(-1+T^2\right)\delta+\hbar\right)^2\left(\eta\xi\hbar+\delta\left(1+y\eta+x\xi\right)\hbar+\delta^2\left(-1+T^2+xy\hbar\right)\right)+\gamma\Big(\eta\xi\hbar^2\left(\left(-1+3T^2\right)\eta\left(\left(-1+T^2\right)\xi-2y\hbar\right)+2x\hbar\left(\xi-3T^2\xi+2y\hbar\right)\right)+\left(-1+T^2\right)\delta^4\left(-2+6T^6-x^2y^2\hbar^2-2T^4(7+4xy\hbar)+T^2\left(10+8xy\hbar-5x^2y^2\hbar^2\right)\right)-4\delta^3\hbar\left(1-3T^6+x^2y^2\hbar^2+T^4\left(7+2xy\left(3+y\eta\right)\hbar+2x^2y\xi\hbar\right)+T^2\left(-5-2xy\left(3+y\eta\right)\hbar+x^2y\hbar\left(-2\xi+y\hbar\right)\right)\right)+2\delta\hbar^2\left(\left(1-3T^2\right)y^2\eta^2\hbar+2\eta\left(\xi+3T^4\xi-4T^2\xi\left(1+xy\hbar\right)+y\hbar\left(1-3T^2+xy\hbar\right)\right)+x\hbar\left(\left(x-3T^2x\right)\xi^2+2y\hbar+\xi\left(2-6T^2+2xy\hbar\right)\right)\right)-\delta^2\hbar\Big(\left(1-4T^2+3T^4\right)y^2\eta^2\hbar+\hbar\Big(-2+3T^4\left(-2+4x\xi+x^2\xi^2\right)+4x(\xi+y\hbar)+x^2\left(\xi^2+2y\xi\hbar-4y^2\hbar^2\right)-2T^2\left(-4+x\left(8\xi-6y\hbar\right)+x^2\xi\left(2\xi-5y\hbar\right)\right)\Big)+2\eta\left(-2\left(-1+T^2\right)\xi\left(1+3T^4-2T^2\left(2+xy\hbar\right)\right)+y\hbar\left(2+6T^4+xy\hbar+T^2\left(-8+5xy\hbar\right)\right)\right)\Big)\Big)\Big)$$

```
{Short[lhs = SimpT@𝕆_QU[SS[e^(ħ (ξ x+η y+δ x y))], {x, y}], 5],
 rhs = SimpT@𝕆_QU[SS[
     e^(ħ ν (ξ x+η y+δ x y-(T^2-1) ξ η)) QΛ[T, y, a, x, ħ ξ, ħ η, ħ δ] /. ν → (1 + (T^2 - 1) δ)^-1], {y, a, x}];
 HL[Simplify[lhs == rhs]]}
```

$$\Big\{\left(1-t\delta\hbar+\left(-\frac{t^2\delta}{2}+t^2\delta^2+t\gamma\delta^2\epsilon-t\eta\xi\right)\hbar^2+\left(-\frac{t^3\delta}{6}+t^3\delta^2-t^3\delta^3+2t^2\gamma\delta^2\epsilon-3t^2\gamma\delta^3\epsilon-\frac{1}{2}t^2\eta\xi+2t^2\delta\eta\xi+2t\gamma\delta\epsilon\eta\xi\right)\hbar^3\right)QU[\,]+\left(2\delta\epsilon\hbar+\left(2t\delta\epsilon-4t\delta^2\epsilon+2\epsilon\eta\xi\right)\hbar^2+\left(t^2\delta\epsilon-6t^2\delta^2\epsilon+6t^2\delta^3\epsilon+2t\epsilon\eta\xi-8t\delta\epsilon\eta\xi\right)\hbar^3\right)QU[a]+\ll 21\gg+\frac{1}{2}\delta^2\eta\hbar^3\,QU[y,y,y,x,x]+\frac{1}{6}\delta^3\hbar^3\,QU[y,y,y,x,x,x],\ \text{True}\Big\}$$

C𝕆, Q𝕆, and Swaps

Swaps from Pensieve://Talks/Toulouse-1705/DogmaDemo.nb and from Pensieve://Talks/Sydney-1708/ExtraDetails@@.nb.

CdsO

```
SetAttributes[{C𝕆, Q𝕆}, Orderless];
CU@C𝕆[specs___, 𝔼[L_, Q_, P_]] := 𝕆_CU[SS[e^(L+Q) P], specs];
QU@Q𝕆[specs___, 𝔼[L_, Q_, P_]] := 𝕆_QU[SS[e^(L+Q) P], specs];
```

```
CU@C𝕆[𝔼[ħ t1 a2, ħ t1^-1 (e^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}_1, {x2, a2, y2}_2] // Short
```

$$CU[\,]+\ll 27\gg+CU[y_1,x_1]\left(-\gamma\epsilon\hbar^2t_2+e^{t_1}\gamma\epsilon\hbar^2t_2+\frac{\epsilon\hbar t_2}{t_1}-\frac{e^{t_1}\epsilon\hbar t_2}{t_1}+\frac{1}{2}\gamma^2\epsilon\hbar^3t_1t_2-\frac{1}{2}e^{t_1}\gamma^2\epsilon\hbar^3t_1t_2\right)$$

```
HL[ρ[e^(ξ CU@x)].ρ[e^(α CU@a)] == ρ[e^(α CU@a)].ρ[e^(e^(-γ α) ξ CU@x)]]
```

True

SW

```
SW_{x_i_, a_j_}[(𝒪 : C𝕆 | Q𝕆)[OrderlessPatternSequence[{lh___, x_i_, a_j_, rh___}_s_,
    more___, 𝔼[L_, Q_, P_]]]] := 𝒪[{lh, a_j, x_i, rh}_s, more,
  With[{q = ⅇ^(-γ α) ξ x_i + α a_j},
   𝔼[L, ⅇ^(-γ α) ξ x_i + (Q /. x_i → 0), ⅇ^-q DP_{x_i→D_ξ, a_j→D_α}[P][ⅇ^q]] /. {α → ∂_{a_j} L, ξ → ∂_{x_i} Q}]]
```

```
c𝕆 = C𝕆[𝔼[ℏ t_1 a_2, ℏ t_1^-1 (ⅇ^t_1 - 1) y_1 x_2, 1 + ϵ x_1 y_2], {y_1, x_1}_1, {x_2, a_2, y_2}_2]
```

$$C\mathbb{O}\left[\{y_1, x_1\}_1, \{x_2, a_2, y_2\}_2, \mathbb{E}\left[\hbar\, a_2\, t_1, \frac{(-1 + e^{t_1})\, \hbar\, x_2\, y_1}{t_1}, 1 + \epsilon\, x_1\, y_2\right]\right]$$

```
SW_{x_2, a_2}[c𝕆]
```

$$C\mathbb{O}\left[\{y_1, x_1\}_1, \{a_2, x_2, y_2\}_2, \mathbb{E}\left[\hbar\, a_2\, t_1, \frac{e^{-\gamma\hbar t_1}(-1 + e^{t_1})\, \hbar\, x_2\, y_1}{t_1}, 1 + \epsilon\, x_1\, y_2\right]\right]$$

```
With[{c𝕆 = C𝕆[{y_1, x_1}_1, {x_2, a_2, y_2}_2, 𝔼[ℏ t_1 a_2, ℏ t_1^-1 (ⅇ^t_1 - 1) y_1 x_2, 1 + ϵ x_1 y_2]]},
 HL[CU[c𝕆] == CU[c𝕆 // SW_{x_2, a_2}]]]
```

True

```
With[{c𝕆 = C𝕆[{y_1, a_1, x_1}_1, {x_2, a_2, y_2}_2,
    𝔼[ℏ (l_11 t_1 a_1 + l_12 t_1 a_2 + l_21 t_2 a_1 + l_22 t_2 a_2), ℏ (γ_11 x_1 y_1 + γ_12 x_1 y_2 + γ_21 x_2 y_1 + γ_22 x_2 y_2),
     1 + ϵ (l_1 a_1 + l_2 a_2 + p_11 x_1 y_1 + p_12 x_1 y_2 + p_21 x_2 y_1 + p_22 x_2 y_2)]]},
 {CU[c𝕆] // Short, HL[CU[c𝕆] == CU[c𝕆 // SW_{x_2, a_2}]]}
]
```

{ <<1>> , True}

```
With[{q𝕆 = Q𝕆[{y_1, a_1, x_1}_1, {x_2, a_2, y_2}_2,
    𝔼[ℏ (l_11 t_1 a_1 + l_12 t_1 a_2 + l_21 t_2 a_1 + l_22 t_2 a_2), ℏ (γ_11 x_1 y_1 + γ_12 x_1 y_2 + γ_21 x_2 y_1 + γ_22 x_2 y_2),
     1 + ϵ (l_1 a_1 + l_2 a_2 + p_11 x_1 y_1 + p_12 x_1 y_2 + p_21 x_2 y_1 + p_22 x_2 y_2)]]},
 {QU[q𝕆] // Short, HL[QU[q𝕆] == QU[q𝕆 // SW_{x_2, a_2}]]}
]
```

{ <<1>> , True}

SW

```
SW_{a_j_, y_i_}[(𝒪 : C𝕆 | Q𝕆)[OrderlessPatternSequence[{lh___, a_j_, y_i_, rh___}_s_,
    more___, 𝔼[L_, Q_, P_]]]] := 𝒪[{lh, y_i, a_j, rh}_s, more,
  With[{q = ⅇ^(-γ α) η y_i + α a_j},
   𝔼[L, ⅇ^(-γ α) η y_i + (Q /. y_i → 0), ⅇ^-q DP_{y_i→D_η, a_j→D_α}[P][ⅇ^q]] /. {α → ∂_{a_j} L, η → ∂_{y_i} Q}]]
```

```
With[{q𝕆 = Q𝕆[{y_1, a_1, x_1}_1, {x_2, a_2, y_2}_2,
    𝔼[ℏ (l_11 t_1 a_1 + l_12 t_1 a_2 + l_21 t_2 a_1 + l_22 t_2 a_2), ℏ (γ_11 x_1 y_1 + γ_12 x_1 y_2 + γ_21 x_2 y_1 + γ_22 x_2 y_2),
     1 + ϵ (l_1 a_1 + l_2 a_2 + p_11 x_1 y_1 + p_12 x_1 y_2 + p_21 x_2 y_1 + p_22 x_2 y_2)]]},
 {QU[q𝕆] // Short, HL[QU[q𝕆] == QU[q𝕆 // SW_{a_2, y_2}]]}
]
```

{ <<1>> , True}

SW

```
SW_{x_{i_},y_{j_}→k_}[C𝕆[{lh___, x_{i_}, y_{j_}, rh___}_{s_}, more___, 𝔼[L_, Q_, P_]]] :=
  C𝕆[{lh, y_k, a_k, x_k, rh}_s, more,
   With[{q = ν (ξ x_k + η y_k + δ x_k y_k - t_k ξ η)},
    𝔼[L, q + (Q /. x_i | y_j → 0), e^{-q} DP_{x_i→D_ξ, y_j→D_η}[P][CΛ[t_k, y_k, a_k, x_k, ξ, η, δ] e^q]] /.
     ν → (1 + t_k δ)^{-1} /. {ξ → (∂_{x_i} Q /. y_j → 0), η → (∂_{y_j} Q /. x_i → 0), δ → ∂_{x_i,y_j} Q}]]
```

```
With[{c𝕆 = C𝕆[{x_1, y_1}_1, {x_2, a_2, y_2}_2,
     𝔼[ℏ (l_12 t_1 a_2 + l_22 t_2 a_2), ℏ (γ_11 x_1 y_1 + γ_12 x_1 y_2 + γ_21 x_2 y_1 + γ_22 x_2 y_2),
      1 + ∈ (l_2 a_2 + p_11 x_1 y_1 + p_12 x_1 y_2 + p_21 x_2 y_1 + p_22 x_2 y_2)]]},
 {CU[c𝕆] // Short, HL[CU[c𝕆] == CU[c𝕆 // SW_{x_1,y_1→1}]]}
]
```

$\{12 \in^2 \hbar^3 \mathrm{CU}[y_1, a_1, a_1, x_1]\, \gamma_{11}^3 + \frac{16}{3} \in^2 \hbar^3 \ll 1 \gg p_{11} \gamma_{11}^3 +$
$\ll 157 \gg + \ll 1 \gg + \mathrm{CU}[\,]\,(1 - \in p_{11} t_1 - \in p_{22} t_2 + \ll 294 \gg + 3 \gamma \in \hbar^3 l_2 t_2^3 \gamma_{22}^3 +$
$24 \gamma \in^2 \hbar^3 p_{22} t_2^3 \gamma_{22}^3 + \in \hbar^3 p_{11} t_1 t_2^3 \gamma_{22}^3 + 4 \in \hbar^3 p_{22} t_2^4 \gamma_{22}^3),\ \mathrm{True}\}$

SW

```
SW_{x_{i_},y_{j_}→k_}[Q𝕆[{lh___, x_{i_}, y_{j_}, rh___}_{s_}, more___, 𝔼[L_, Q_, P_]]] :=
  Q𝕆[{lh, y_k, a_k, x_k, rh}_s, more,
   With[{q = ν (ξ x_k + η y_k + δ x_k y_k - ℏ^{-1} (T_k^2 - 1) ξ η)},
    𝔼[L, q + (Q /. x_i | y_j → 0), e^{-q} DP_{x_i→D_ξ, y_j→D_η}[P][QΛ[T_k, y_k, a_k, x_k, ξ, η, δ] e^q]] /.
     ν → (1 + ℏ^{-1} (T_k^2 - 1) δ)^{-1} /. {ξ → (∂_{x_i} Q /. y_j → 0), η → (∂_{y_j} Q /. x_i → 0), δ → ∂_{x_i,y_j} Q}]]
```

```
With[{q𝕆 = Q𝕆[{x_1, y_1}_1, {x_2, a_2, y_2}_2,
     𝔼[ℏ (l_12 t_1 a_2 + l_22 t_2 a_2), ℏ (γ_11 x_1 y_1 + γ_12 x_1 y_2 + γ_21 x_2 y_1 + γ_22 x_2 y_2),
      1 + ∈ (l_2 a_2 + p_11 x_1 y_1 + p_12 x_1 y_2 + p_21 x_2 y_1 + p_22 x_2 y_2)]]},
 {QU[q𝕆] // Short, HL[err = SimpT[QU[q𝕆] - QU[q𝕆 // SW_{x_1,y_1→1}]]]}
]
```

$\{\frac{1}{6} \in \hbar^3 \mathrm{QU}[y_1, y_1, y_1, y_1, x_1, x_1, x_1, x_1]\, p_{11} \gamma_{11}^3 + \ll 159 \gg + \mathrm{QU}[\,]$
$\left(1 + \frac{\in p_{11}}{\hbar} + \frac{\in p_{22}}{\hbar} - \gamma \in l_{12} p_{22} t_1 + \ll 1713 \gg + \frac{\in p_{11} T_1^2 T_2^6 \gamma_{22}^3}{\hbar} + 36 \gamma \in^2 p_{22} T_2^8 \gamma_{22}^3 + \frac{4 \in p_{22} T_2^8 \gamma_{22}^3}{\hbar}\right),\ 0\}$

Stitching Direct

```
MatrixExp[η_1 ρ[CU@y]].MatrixExp[α_1 ρ[CU@a]].MatrixExp[ξ_1 ρ[CU@x]].MatrixExp[η_2 ρ[CU@y]].
  MatrixExp[α_2 ρ[CU@a]].MatrixExp[ξ_2 ρ[CU@x]] // Simplify // MatrixForm
```

$$\begin{pmatrix} e^{\gamma(\alpha_1+\alpha_2)} (1 + \gamma \in \eta_2 \xi_1) & e^{\gamma \alpha_1} \gamma (e^{\gamma \alpha_2} \xi_2 + \xi_1 (1 + e^{\gamma \alpha_2} \gamma \in \eta_2 \xi_2)) \\ e^{\gamma \alpha_2} \in (\eta_2 + e^{\gamma \alpha_1} \eta_1 (1 + \gamma \in \eta_2 \xi_1)) & 1 + e^{\gamma \alpha_1} \gamma \in \eta_1 \xi_1 + e^{\gamma \alpha_2} \gamma \in (\eta_2 + e^{\gamma \alpha_1} \eta_1 (1 + \gamma \in \eta_2 \xi_1)) \xi_2 \end{pmatrix}$$

```
eqn = MatrixExp[η1 ρ[CU@y]].MatrixExp[α1 ρ[CU@a]].MatrixExp[ξ1 ρ[CU@x]].
   MatrixExp[η2 ρ[CU@y]].MatrixExp[α2 ρ[CU@a]].MatrixExp[ξ2 ρ[CU@x]] ==
  e^(τ0 ϵ γ) MatrixExp[η0 ρ[CU@y]].MatrixExp[α0 ρ[CU@a]].MatrixExp[ξ0 ρ[CU@x]]
```

$$\{\{e^{\gamma\alpha_2}(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\epsilon\eta_2\xi_1),\ e^{\gamma\alpha_1}\gamma\xi_1+e^{\gamma\alpha_2}\gamma(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\epsilon\eta_2\xi_1)\xi_2\},\ \{e^{\gamma\alpha_2}(e^{\gamma\alpha_1}\epsilon\eta_1+\epsilon\eta_2(1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1)),\ 1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1+e^{\gamma\alpha_2}\gamma(e^{\gamma\alpha_1}\epsilon\eta_1+\epsilon\eta_2(1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1))\xi_2\}\} == \{\{e^{\alpha0\gamma+\gamma\epsilon\tau0},\ e^{\alpha0\gamma+\gamma\epsilon\tau0}\gamma\xi0\},\ \{e^{\alpha0\gamma+\gamma\epsilon\tau0}\epsilon\eta0,\ e^{\gamma\epsilon\tau0}(1+e^{\alpha0\gamma}\gamma\epsilon\eta0\,\xi0)\}\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {τ0, η0, α0, ξ0}]][[1]]
```

Solve: Inconsistent or redundant transcendental equation. After reduction, the bad equation is

$\text{Log}[e^{\gamma(\alpha0+\epsilon\tau0)}] - \text{Log}[e^{\gamma\alpha_2}(e^{\gamma\,\text{Subscript}[\ll 2\gg]}+e^{\text{Times}[\ll 2\gg]}\gamma\epsilon\eta_2\xi_1)] == 0.$

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Solve: Equations may not give solutions for all "solve" variables.

$$\Big\{\tau0\to\frac{1}{\gamma\epsilon}\left(-\text{Log}[e^{\alpha0\gamma}]+\text{Log}[e^{\gamma\alpha_1+\gamma\alpha_2}+e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_2\xi_1]\right),\ \eta0\to\frac{1}{\gamma\epsilon\left(\xi_1+e^{\gamma\alpha_2}\xi_2+e^{\gamma\alpha_2}\gamma\epsilon\eta_2\xi_1\xi_2\right)}$$

$$e^{-\gamma\alpha_1}\Big(\frac{1}{2}+\frac{1}{2}e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1+\frac{1}{2}e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\xi_2+\frac{1}{2}e^{\gamma\alpha_2}\gamma\epsilon\eta_2\xi_2+\frac{1}{2}e^{\gamma\alpha_1+\gamma\alpha_2}\gamma^2\epsilon^2\eta_1\eta_2\xi_1\xi_2-$$

$$\frac{1}{2}\sqrt{\Big((-1-e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\xi_2-e^{\gamma\alpha_2}\gamma\epsilon\eta_2\xi_2-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma^2\epsilon^2\eta_1\eta_2\xi_1\xi_2)^2+}$$

$$4e^{-\alpha0\gamma+\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\,(-e^{\gamma\alpha_1}\eta_1\xi_1-\eta_2\xi_1-e^{\gamma\alpha_1}\gamma\epsilon\eta_1\eta_2\xi_1^2-e^{\gamma\alpha_1+\gamma\alpha_2}\eta_1\xi_2-e^{\gamma\alpha_2}\eta_2\xi_2-$$

$$2e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\eta_2\xi_1\xi_2-e^{\gamma\alpha_2}\gamma\epsilon\eta_2^2\xi_1\xi_2-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma^2\epsilon^2\eta_1\eta_2^2\xi_1^2\xi_2)\Big)\Big),$$

$$\xi0\to\frac{1}{e^{\gamma\alpha_1}\eta_1+\eta_2+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\eta_2\xi_1}\,e^{-\gamma\alpha_2}\Big(\frac{1}{2\gamma\epsilon}+\frac{1}{2}e^{\gamma\alpha_1}\eta_1\xi_1+\frac{1}{2}e^{\gamma\alpha_1+\gamma\alpha_2}\eta_1\xi_2+$$

$$\frac{1}{2}e^{\gamma\alpha_2}\eta_2\xi_2+\frac{1}{2}e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\eta_2\xi_1\xi_2-$$

$$\frac{1}{2\gamma\epsilon}\Big(\sqrt{\Big((-1-e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\xi_2-e^{\gamma\alpha_2}\gamma\epsilon\eta_2\xi_2-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma^2\epsilon^2\eta_1\eta_2\xi_1\xi_2)^2+}$$

$$4e^{-\alpha0\gamma+\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\,(-e^{\gamma\alpha_1}\eta_1\xi_1-\eta_2\xi_1-e^{\gamma\alpha_1}\gamma\epsilon\eta_1\eta_2\xi_1^2-e^{\gamma\alpha_1+\gamma\alpha_2}\eta_1\xi_2-e^{\gamma\alpha_2}\eta_2\xi_2-$$

$$2e^{\gamma\alpha_1+\gamma\alpha_2}\gamma\epsilon\eta_1\eta_2\xi_1\xi_2-e^{\gamma\alpha_2}\gamma\epsilon\eta_2^2\xi_1\xi_2-e^{\gamma\alpha_1+\gamma\alpha_2}\gamma^2\epsilon^2\eta_1\eta_2^2\xi_1^2\xi_2)\Big)\Big)\Big)\Big\}$$

```
eqn = MatrixExp[η1 ρ[CU@y]].MatrixExp[α1 ρ[CU@a]].MatrixExp[ξ1 ρ[CU@x]].
   MatrixExp[η2 ρ[CU@y]].MatrixExp[α2 ρ[CU@a]].MatrixExp[ξ2 ρ[CU@x]] ==
  T0 MatrixExp[η0 ρ[CU@y]].MatrixExp[α0 ρ[CU@a]].MatrixExp[ξ0 ρ[CU@x]]
```

$$\{\{e^{\gamma\alpha_2}(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\epsilon\eta_2\xi_1),\ e^{\gamma\alpha_1}\gamma\xi_1+e^{\gamma\alpha_2}\gamma(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\epsilon\eta_2\xi_1)\xi_2\},\ \{e^{\gamma\alpha_2}(e^{\gamma\alpha_1}\epsilon\eta_1+\epsilon\eta_2(1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1)),\ 1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1+e^{\gamma\alpha_2}\gamma(e^{\gamma\alpha_1}\epsilon\eta_1+\epsilon\eta_2(1+e^{\gamma\alpha_1}\gamma\epsilon\eta_1\xi_1))\xi_2\}\} == \{\{e^{\alpha0\gamma}T0,\ e^{\alpha0\gamma}T0\,\gamma\xi0\},\ \{e^{\alpha0\gamma}T0\,\epsilon\eta0,\ T0(1+e^{\alpha0\gamma}\gamma\epsilon\eta0\,\xi0)\}\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {T0, η0, α0, ξ0}]][[1]]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

$$\left\{T0 \to \frac{1}{1+\gamma \epsilon \eta_2 \xi_1},\ \eta 0 \to \frac{\eta_1 + e^{-\gamma \alpha_1} \eta_2 + \gamma \epsilon \eta_1 \eta_2 \xi_1}{1+\gamma \epsilon \eta_2 \xi_1},\right.$$
$$\left.\alpha 0 \to \frac{\mathrm{Log}\left[e^{\gamma \alpha_1 + \gamma \alpha_2}\left(1+\gamma \epsilon \eta_2 \xi_1\right)^2\right]}{\gamma},\ \xi 0 \to \frac{e^{-\gamma \alpha_2} \xi_1 + \xi_2 + \gamma \epsilon \eta_2 \xi_1 \xi_2}{1+\gamma \epsilon \eta_2 \xi_1}\right\}$$

𝔼

𝔼[L, Q, P] means $e^{\hbar(L+Q)} P$, where L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial. It should be interpreted via C𝕆[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...] (with some default for direct interpretation), or likewise via Q𝕆[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...]. In themselves, C𝕆 and Q𝕆 should have an interpretation in CU/QU by casting.